Pensieve header: Implementing Θ - the main notebook accompanying Talks/KnotTheory-Congress-2502

tex

\begin{frame}

pdf

Invariance under R3

pdf

This is Theta.nb of http://drorbn.net/ktc25/ap.

```
In[∘]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Talks\\KnotTheoryCongress-2502"];
```

pdf

```
In[∘]:= Once[<< KnotTheory`; << Rot.m; << PolyPlot.m];
```

pdf

```
Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
Loading Rot.m from http://drorbn.net/ktc25/ap to compute rotation numbers.
```

pdf

```
Loading PolyPlot.m from http://drorbn.net/ktc25/ap to plot 2-variable polynomials.
```

pdf

```
In[∘]:= T_3 = T_1 T_2;
```

pdf

```
In[∘]:= CF[ℰ_] := Expand@Collect[ℰ, g__, F] /. F → Factor;
```

tex

\end{frame} \begin{frame}

pdf

```
In[∘]:= F_1[{s_, i_, j_}] =
   CF[s (1 / 2 - g_3ii + T_2^s g_1ii g_2ji - g_1ii g_2jj - (T_2^s - 1) g_2ji g_3ii + 2 g_2jj g_3ii - (1 - T_3^s) g_2ji g_3ji -
       g_2ii g_3jj - T_2^s g_2ji g_3jj + g_1ii g_3jj + ((T_1^s - 1) g_1ji (T_2^(2 s) g_2ji - T_2^s g_2jj + T_2^s g_3jj) +
          (T_3^s - 1) g_3ji (1 - T_2^s g_1ii - (T_1^s - 1) (T_2^s + 1) g_1ji + (T_2^s - 2) g_2jj + g_2ij)) / (T_2^s - 1))];
```

pdf

```
In[∘]:= F_2[{s0_, i0_, j0_}, {s1_, i1_, j1_}] := CF[s1 (T_1^s0 - 1) (T_2^s1 - 1)^-1
     (T_3^s1 - 1) g_{1,j1,i0} g_{3,j0,i1} ( (T_2^s0 g_{2,i1,i0} - g_{2,i1,j0}) - (T_2^s0 g_{2,j1,i0} - g_{2,j1,j0}))]
```

pdf

```
In[∘]:= F_3[φ_, k_] = -φ / 2 + φ g_3kk;
```

tex

\end{frame} \begin{frame}

pdf

In[∘]:=
```
δ_{i_,j_} := If[i === j, 1, 0];
gR_{s_,i_,j_} := {
  g_{ν_ jβ_} :→ g_{ν j⁺ β} + δ_{jβ}, g_{ν_ iβ_} :→ T_ν^s g_{ν i⁺ β} + (1 - T_ν^s) g_{ν j⁺ β} + δ_{iβ},
  g_{ν_ α_ i⁺} :→ T_ν^s g_{ναi} + δ_{αi⁺}, g_{ν_ α_ j⁺} :→ g_{ναj} + (1 - T_ν^s) g_{ναi} + δ_{αj⁺}
 }
```

Proof of Reidemeister 3:

tex

\end{frame} \begin{frame}
\[\resizebox{0.75\linewidth}{!}{\import{../Toronto-241030}{R3.pdftex_t}} \]

pdf

```
DSum[Cs___] := Sum[F_1[c], {c, {Cs}}] + Sum[F_2[c0, c1], {c0, {Cs}}, {c1, {Cs}}]
lhs = DSum[{1, j, k}, {1, i, k⁺}, {1, i⁺, j⁺}, {s, m, n}] //. gR_{1,j,k} ⋃ gR_{1,i,k⁺} ⋃ gR_{1,i⁺,j⁺};
rhs = DSum[{1, i, j}, {1, i⁺, k}, {1, j⁺, k⁺}, {s, m, n}] //. gR_{1,i,j} ⋃ gR_{1,i⁺,k} ⋃ gR_{1,j⁺,k⁺};
Simplify[lhs == rhs]
```

Out[∘]=
pdf

```
True
```

tex

\end{frame} \begin{frame}

pdf

The Main Program

pdf

In[∘]:=
```
Θ[K_] := Module[{Cs, φ, n, A, Δ, G, ev, θ},
  {Cs, φ} = Rot[K]; n = Length[Cs];
  A = IdentityMatrix[2 n + 1];
  Cases[Cs, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1
                                                             0     -1    ))];
  Δ = T^((-Total[φ] - Total[Cs⟦All,1⟧])/2) Det[A];
  G = Inverse[A];
  ev[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :→ (G⟦α, β⟧ /. T → T_ν)];
  θ = ev[∑_{k=1}^{n} F_1[Cs⟦k⟧]];
  θ += ev[∑_{k1=1}^{n} ∑_{k2=1}^{n} F_2[Cs⟦k1⟧, Cs⟦k2⟧]];

  θ += ev[∑_{k=1}^{2 n} F_3[φ⟦k⟧, k]];
  Factor@{Δ, (Δ /. T → T_1) (Δ /. T → T_2) (Δ /. T → T_3) θ}];
```

tex

\end{frame}

The Rolfsen Table

In[∘]:=
```
tab250 = {{1, 0}}~Join~Table[Θ[K], {K, AllKnots[{3, 10}]}];
```

KnotTheory: Loading precomputed data in PD4Knots`.

In[∘]:=
```
g250 = GraphicsGrid[Partition[PolyPlot /@ tab250, 25], Spacings → 0, Dividers → All]
```

Out[∘]=



In[∘]:=
```
Export["g250.png", g250]
```

Out[∘]=

g250.png

tex

\begin{frame}

pdf

The Trefoil

pdf

In[∘]:=
```
Θ[Knot[3, 1]] // Expand
```

Out[∘]=
pdf

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2\,T_2^2}+\frac{1}{T_1\,T_2^2}+\frac{1}{T_1^2\,T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2\,T_2-T_2^2+T_1\,T_2^2-T_1^2\,T_2^2\right\}$$

pdf

In[ ]:=
```
PolyPlot[Θ[Knot[3, 1]], ImageSize → Tiny]
```

Out[ ]=
pdf



tex

\end{frame} \begin{frame}

pdf

The Conway and Kinoshita-Terasaka Knots

tex

\[\includegraphics[width=15mm]{../Toronto-241030/K11n34.png}
\qquad
\includegraphics[width=15mm]{../Toronto-241030/K11n42.png}
\]



pdf

```
GraphicsRow[PolyPlot[Θ[Knot[#]], ImageSize → Tiny] & /@ {"K11n34", "K11n42"}]
```

Out[ ]=
pdf



tex

(Note that the genus of the Conway knot appears to be bigger than the genus of Kinoshita-Terasaka)
\end{frame}
\begin{frame}

pdf

The Torus Knots $T_{13/2}$, $T_{17/3}$, $T_{13,5}$, and $T_{7,6}$

pdf

In[∘]:=
```
GraphicsRow[ImageCompose[
    PolyPlot[Θ[TorusKnot @@ #], ImageSize → 480],
    TubePlot[TorusKnot @@ #, ImageSize → 240],
    {Right, Bottom}, {Right, Bottom}
   ] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}]
```

Out[∘]=
pdf



tex

\end{frame}

The 132-crossing Torus Knot $T_{22/7}$

In[∘]:=
```
AbsoluteTiming[T227 = Θ[TorusKnot[22, 7]];]
```

Out[∘]=

```
{700.773, Null}
```

In[∘]:=
```
T227Plot = ImageCompose[
  PolyPlot[T227],
  TubePlot[TorusKnot[22, 7], ImageSize → 240],
  {Right, Bottom}, {Right, Bottom}
 ]
```

Out[∘]=



In[∘]:=
```
Export["T227Plot.pdf", T227Plot]
```

Out[∘]=

T227Plot.pdf

Large Random Knots

See also https://drorbn.net/AcademicPensieve/Projects/HigherRank/DunfieldKnots/.

In[∘]:=
```
DK300 = PolyPlot@Expand[
   Get["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank\\DunfieldKnots\\D300.m"][[
     2]] /. {T1 → T₁, T2 → T₂}]
```

Out[∘]=



In[∘]:=
```
Export["DK300.png", DK300, ImagePadding → None, PlotRangePadding → None]
```

Out[∘]=
```
DK300.png
```

In[∘]:=
```
Export["DK300.pdf", DK300]
```

Out[∘]=
```
DK300.pdf
```

Some Virtual Knots

In[ ]:=
```
pd = PD[X[2, 4, 3, 1], X[3, 1, 4, 2]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[ ]=

{{{-1, 4, 2}, {-1, 1, 3}}, {0, 0, 1, 0}}

Out[ ]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{-1 - 2 T_2 + 4 T_1 T_2}{2 T_1 T_2}\right\}$$

Out[ ]=



In[∘]:=
```
pd = PD[X[3, 8, 4, 1], X[1, 6, 2, 7], X[5, 2, 6, 3], X[7, 5, 8, 4]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[∘]=

{{{-1, 8, 3}, {-1, 6, 1}, {-1, 2, 5}, {1, 4, 7}}, {0, 0, 0, 0, 1, -1, -1, -1}}

Out[∘]=

$$\left\{-\frac{1-3T+T^2}{T},\ -\frac{\left(1-3T_1+T_1^2\right)\left(-1+T_1+2T_1^2+T_2-7T_1^2T_2-2T_1^3T_2-T_1T_2^2+4T_1^2T_2^2+5T_1^3T_2^2-3T_1^2T_2^3+T_1^3T_2^3\right)}{T_1^2}\right\}$$

Out[∘]=



In[∘]:=
```
pd = PD[X[2, 8, 3, 1], X[3, 1, 4, 2], X[6, 4, 7, 5], X[7, 5, 8, 6]];
Rot[pd]
th = PowerExpand[Θ[pd]]
PolyPlot[th]
```

Out[∘]=

{{{-1, 8, 2}, {-1, 1, 3}, {-1, 4, 6}, {-1, 5, 7}}, {0, 0, 1, 0, 0, 1, 1, 0}}

Out[∘]=

$$\left\{\frac{1}{\sqrt{T}}, \frac{1}{2 T_1^3 T_2^4}\left(4 - 4 T_1 - 6 T_2 + 2 T_1 T_2 + 6 T_1^2 T_2 + 4 T_1 T_2^2 - 8 T_1^2 T_2^2 + 2 T_2^3 + 2 T_1 T_2^3 + T_1^2 T_2^3 - 4 T_1^3 T_2^3 - 2 T_1 T_2^4 - 6 T_1^2 T_2^4 + 10 T_1^3 T_2^4\right)\right\}$$

Out[∘]=